
Abstract—Sketch maps are an externalization of an indi-
vidual’s mental images of an environment. The information
represented in sketch maps is schematized, distorted, generalized,
and thus processing spatial information in sketch maps requires
plausible representations based on human cognition. Typically
only qualitative relations between spatial objects are preserved
in sketch maps, and therefore processing spatial information on
a qualitative level has been suggested. This study extends our
previous work on qualitative representations and alignment of
sketch maps. In this study, we define a set of spatial relations
using the declarative spatial reasoning system CLP(QS) as an
approach to formalizing key spatial aspects that are preserved
in sketch maps. Unlike geo-referenced maps, sketch maps do
not have a single, global reference frame. Rather, the sketched
elements themselves act as referencing objects. Using the declara-
tive spatial reasoning system CLP(QS), we define constraint logic
programming rules that formalize various key spatial aspects of
sketch maps at a local level, between nearby objects. These rules
focus on linear ordering, cyclic ordering, and relative orientation
of depicted objects along, and around, salient reference objects.

I. INTRODUCTION

S
KETCH maps are used to externalize an individual’s men-

tal image of the environment. The information represented

in sketch maps is based on observation rather than measure-

ments. Therefore, information in sketch maps is schematized,

distorted, and generalized. Freehand sketch maps contain

objects and spatial relations between these objects which

enable users to use sketch maps to communicate and reason

about actions in environments. During the last two decades,

several approaches [7, 9, 21, 32] attempt to capture spatial

configurations between depicted objects qualitatively. These

approaches use different aspects of space such as topological

relations, cardinal directions, relative orientations, and relative

distances.

Throughout a series of experiments [27, 33, 34], Wang

et al. identify a set of sketch aspects which are not subject

to schematizations, distortions or any other cognitive impact.

These sketch aspects represent: linear ordering, cyclic order-

ing, relative orientations, and topological relations between

spatial objects. The identified sketch aspects are categorized

into local and global levels [27]. The local level relations

refer to the relations between nearby objects while global level

relations represent possible relations between all objects in a

map. During the last two decades, a series of qualitative spatial

calculi have been proposed in the area of Qualitative Spatial

Reasoning (QSR) [11] to formalize some of these aspects

such as representations for the topological relations [6, 23],

orderings [1, 22, 25], directions [10, 24], relative position of

points [19, 20, 24] and others.

In our previous studies [14, 15, 16, 27], we propose a

set of plausible representations and their coarse versions to

qualitatively formalize key sketch aspects. We use several

qualifiers to extract qualitative constraints from geometric

representations of sketch and geo-referenced maps [13] in

the form of Qualitative Constraint Networks (QCNs). QCNs

are complete graphs representing spatial objects and relations

between them. However, in order to derive more cognitively

accurate QCNs, we require greater flexibility in being able to

define qualitative spatial relations for our particular application

domain, i.e. geographic-scale sketch maps. Specifically, sketch

maps require qualitative constraints at a local level between

particular types of adjacent objects such as linear ordering,

cyclic ordering, and orientation information of nearby land-

marks with respect to reference objects.

In this study, we propose the utilisation of the declarative

spatial reasoning system CLP(QS) [2, 26] as an alternative

approach to deriving cognitively plausible constraints between

nearby objects. The system is capable of modeling and reason-

ing about qualitative spatial relations within the context of the

constraint logic programming. Using the CLP(QS) framework,

we define logic programming rules over qualitative spatial

domains in order to express and solve declarative, high-level

constraints between spatial objects depicted in sketch maps.

The remainder of this paper is structured as follows: In the

following section, we briefly introduce related work. In Section

3 we discuss spatial objects and cognitively plausible aspects

found in sketch maps. In Section 4 we present CLP(QS) rules

that formalize the cognitively salient aspects of sketch maps.

Section 5 concludes the paper with an outlook on future work.

II. RELATED WORK

The information in sketch maps is based on observations

rather than measurements. Therefore, processing spatial in-

formation on a qualitative level has been suggested [5, 27].

During the last two decades, several approaches [7, 9, 21, 32]

attempt to capture spatial configurations between depicted

objects qualitatively. Egenhofer et al. [8] propose Spatial-

Query-by-Sketch, a sketch-based user interface that focuses

on enabling a user to specify spatial relations (topology and

cardinal directions) by drawing them. Volker et al. [31] pro-

pose the visual query system VISCO. It offers a sketch-based

query language for defining approximate spatial constellations

of the objects. Forbus et al. [9] develop a sketch understanding

system CogSkech which is a space search system that focuses

on topological relations among sketched elements, and reasons

about these relations to infer new knowledge. Nedas et al. [21]

propose a similarity measure methodology for comparing two

spatial scenes by identifying cognitively-motivated similarities

between objects, relations among spatial objects, and the ratio

of the total number of objects in both scenes to the number

of objects that have been correctly matched.

These approaches share our motivation of using abstract

qualitative relations to represent spatial configurations between

objects depicted by a user. In a previous study [27], we propose

a framework to preprocess, align and integrate sketched spatial

information on a qualitative level. The framework addresses

the extraction of objects from sketch maps, computing QCNs

from geometric representations of sketch and geo-referenced

maps, and aligning them qualitatively.

This study extends our previous work on qualitative repre-

sentations of spatial objects. As the outline of spatial objects

in freehand sketches are imprecise, the qualitative represen-

tation of spatial objects with imprecise boundaries leads to

different qualitative relations when compared to relations in

geo-referenced maps. In this study, we present spatial rules

defined using the CLP(QS) system as an alternative approach

to compute qualitative relations (on a conceptual level) which

are preserved in freehand sketch maps. These rules address

linear ordering, cyclic ordering, and orientation information

of adjacent objects along and around key reference objects.

III. SPATIAL OBJECTS AND THEIR CONFIGURATIONS IN

SKETCH MAPS

A. Spatial objects

Inspired by the Lynch’s seminal work [18] and Tversky’s

analysis of mental structures [30], we characterize the depicted

objects in sketch maps into four elements: streets segments,

junctions, landmarks, and city-blocks. These elements are

automatically extracted using the object recognition method

proposed in [4].

Street segments are connected, and mostly linear, features

in sketch maps. They are represented as line segments and

are connected to other street segments at junctions. The con-

nectivity of street segments and the street-network is central

for human path planning [12]. Junctions are the end-points

of street segments. The end-points, where street segments are

not connected to other street segments, are called hanging

end-points. At the boundary of the sketching medium, street

segments are left unconnected to any further street segment,

resulting in hanging end-points. In sketch maps, junctions

capture the connectivity of various street segments forming

a street network. In our approach, both hanging end-points

and junctions are spatially represented as 2D points.

Landmarks are the most salient elements in an environment

and are therefore essential to characterize an environment. In

freehand sketches, landmarks are vectorized and approximated

by polygons which represent spatial entities such as water

bodies, buildings, and parks. Landmarks and road entities are

the most frequently depicted spatial objects in sketch maps [3],

while city-blocks are the smallest regions. They are delimited

by a lineal representation of connected street segments. People

do not always sketch complete city-blocks, in particular at the

edge of the sketch medium. Therefore, we define city-blocks

as areas either bounded by the street segments, or bounded by

street segments and the boundary of the medium [14].

B. Invariant spatial aspects in Sketch Maps

Processing sketch information at a qualitative level requires

explicit knowledge about certain aspects of sketches that

are not subject to schematizations, distortions, or any other

cognitive impact [29]. That is, these aspects are preserved

in freehand sketches. Throughout a series of experiments

[27, 33, 34], Wang et al. identified a set of seven invariant

sketch aspects. These aspects consist of: linear ordering of

landmarks and street segments along a route, cyclic ordering of

landmarks and street segments around reference junctions, rel-

ative orientation of landmarks with respect to street segments,

topological and orientation relations between street segments

in street network, and topological relations between extended

objects (landmarks, and city-blocks). This paper focuses on the

formalization of ordering and relative orientation of adjacent

objects using spatial rules defined in the context of the

CLP(QS) system.

IV. SPATIAL RULES FOR QUALITATIVELY EQUIVALENT

CONFIGURATIONS

Using the CLP(QS) framework, we define spatial rules to

compute qualitative information between nearby objects. For

the linear ordering and orientation information of adjacent

landmarks, we use connected street segments as reference

objects, while junctions are used as reference objects for cyclic

ordering. The adjacency of landmarks is defined via relative

metric distances.

A. Preliminaries

CLP(QS) includes a library of qualitative spatial relations

encoded as polynomial constraints over a set of real variables

X , which are solved via constraint logic programming [2].

In this subsection we present the CLP(QS) library implemen-

tations of projection, distance, and orientation relations that

we build on in subsequent sections. A set of spatial relations

is consistent in CLP(QS) if there exists some assignment of

reals to the variables X such that all of the corresponding

polynomial constraints are satisfied. CLP(QS) uses a variety of

polynomial solvers including CLP(R), SAT Modulo Theories,

quantifier elimination by Cylindrical Algebraic Decomposi-

tion, and geometric constraint solvers.

Projection. A point is projected onto a line using the dot

product. This is extended to segment-line projection by pro-

jecting both end points. Polygons are projected onto lines by

projecting all vertices and taking the maximum and minimum

projected values as the projected interval. Points are projected

onto segments by clamping the projected value to lie within

the projection of the segment and the line collinear with the

segment, i.e. let v, a, b be reals such that a ≤ b then

CLAMP(v, a, b) =











v if a ≤ v ≤ b

a if v < a

b if v > b

CLP(QS) projection predicates are implemented as follows.

projection(value(V), point(Xp,Yp),

line(point(Xa,Ya),point(Xb,Yb)) ) :-

{(Xp - Xa) * (Xb - Xa) + (Yp - Ya) * (Yb - Ya) =:= V}.

projection(interval(Prj1,Prj2),

segment(Pa1,Pa2), segment(Pb1,Pb2)) :-

projection(value(Prj1), Pa1, segment(Pb1,Pb2) ),

projection(value(Prj2), Pa2, segment(Pb1,Pb2) ).

projection(value(V), point(Xp,Yp),

segment(point(Xa,Ya),point(Xb,Yb)) ) :-

{(Xp - Xa) * (Xb - Xa) + (Yp - Ya) * (Yb - Ya) =:= PrjV},

{Max =:= (Xb - Xa)^2 + (Yb - Ya)^2},

clpqs_utils:clamp_(0, Max, PrjV, V).

The CLP(QS) predicate for obtaining the coordinates of the

projection onto a segment (or line) is:

reconstruct_projection(point(Xp,Yp), value(V),

segment(point(Xa,Ya),point(Xb,Yb))) :-

{Limit =:= (Xb - Xa)^2 + (Yb - Ya)^2},

{Limit > 0},

{NormV =:= V / Limit},

{Xp =:= (Xb - Xa) * NormV + Xa},

{Yp =:= (Yb - Ya) * NormV + Ya}.

Euclidean distance. We employ CLP(QS) Euclidean distances

between points, and between points and segments.

distance(value(V), point(Xa,Ya), point(Xb,Yb)) :-

{(Xa - Xb)^2 + (Ya - Yb)^2 =:= V^2, V > 0}.

distance(nearer_than,

point(Xa,Ya), point(Xb,Yb), point(Xr,Yr)) :-

{(Xa - Xr)^2 + (Ya - Yr)^2 < (Xb - Xr)^2 + (Yb - Yr)^2}.

distance(equidistant,

point(Xa,Ya), point(Xb,Yb), point(Xr,Yr)) :-

{(Xa - Xr)^2 + (Ya - Yr)^2 =:= (Xb - Xr)^2 + (Yb - Yr)^2}.

distance(farther_than,

point(Xa,Ya), point(Xb,Yb), point(Xr,Yr)) :-

{(Xa - Xr)^2 + (Ya - Yr)^2 > (Xb - Xr)^2 + (Yb - Yr)^2}.

distance(value(V), point(Xp,Yp),

segment(point(Xa,Ya), point(Xb,Yb)) ) :-

projection(point(Xpj, Ypj), point(Xp,Yp),

segment(point(Xa,Ya),point(Xb,Yb)) ),

distance(value(V), point(Xpj, Ypj), point(Xp,Yp)).

Relative Orientation. We employ CLP(QS) relative orienta-

tion predicates between points and lines.

orientation(left_of, point(Xp,Yp),

line(point(Xa,Ya), point(Xb,Yb)) ) :-

{(Xb - Xa) * (Yp - Ya) > (Yb - Ya) * (Xp - Xa)}.

a

b

c

Fig. 1. Inconsistent relative orientation relations between triangle edges
A,B,C and a point D: the point D must occupy the intersection of the red
regions. As the red regions are disconnected, the constraints are unsatisfiable.

orientation(collinear, point(Xp,Yp),

line(point(Xa,Ya), point(Xb,Yb)) ) :-

{(Xb - Xa) * (Yp - Ya) =:= (Yb - Ya) * (Xp - Xa)}.

orientation(right_of, point(Xp,Yp),

line(point(Xa,Ya), point(Xb,Yb)) ) :-

{(Xb - Xa) * (Yp - Ya) < (Yb - Ya) * (Xp - Xa)}.

Spatial reasoning with incomplete numerical information.

We emphasise that, using CLP(QS), it is possible to reason

about sets of spatial relations in the partial or complete absence

of numerical information. For example, we can define an anti-

clockwise triangle with vertices A,B,C with the constraint

that C is on the left of line (A,B). A point D can then be

constrained by relative orientation relations with the triangle

edges:

?- orientation(left_of,C,line(A,B)),

| orientation(left_of,D,line(A,C)),

| orientation(right_of,D,line(A,B)),

| orientation(right_of,D,line(B,C)).

false.

...

| orientation(left_of,D,line(B,C)).

true.

Even though we have not provided any numerical informa-

tion about the positions of the points A,B,C,D, CLP(QS)

correctly determines that D cannot be simultaneously left

of (A,C) and right of (A,B) and (B,C) as illustrated in

Figure 1. We use this feature to reason about possible scenarios

in sketch maps in cases where only incomplete numerical

information is available.

B. Rules for Linear Ordering as Constraints

In sketch maps, linear ordering of spatial objects is an

invariant sketch aspect [33]. It describes the linear ordering of

adjacent landmarks and street segments along a route. A route

is defined as a set of connected street segments. In [33, 34]

Wang et al. found the linear ordering of both landmarks and

street segments as a suitable representation for sketch map

alignment. In our previous studies [16], we proposed a coarse

version of Allen’s interval algebra [1] to extract the linear

ordering of spatial objects along a route.

We define spatial rules to compute the linear ordering of

adjacent landmarks and connected street segments along a

Fig. 2. (a) Spatial objects in the sketch map (b) Ordering of adjacent
landmarks and connecting street segments along the route (AB,BC).

route. The adjacency of landmarks is computed using the rela-

tive metric distance between street segments and landmarks. A

landmark is considered adjacent or local if its footprint inter-

sects with the buffer around the reference street segment. Next,

adjacent landmarks are projected from both sides of a route

onto the street segments. As proposed in our previous studies

[16], the ordering between intervals of projected landmarks

and street segments are represented using both Allen’s Interval

relations and a coarse version of Allen’s interval relations [1].

This is accomplished by encoding qualitative spatial relations

as rules and facts using CLP(QS), as illustrated in Figure 2.

Figure 2b shows the linear ordering of the depicted objects in

terms of Allent’s relations such as meets(m), before(<), and

during(d).

ordering(Rel, St,Lm) :-

object(St, type(street)),

object(Lm, type(landmark)),

relation(proximity(adjacent,Lm,St)),

oriented_representation(St,segment(Pa,Pb)),

representation(Lm,GLm),

projection(PrjLm,GLm,line(Pa,Pb)),

projection(PrjSt,segment(Pa,Pb),line(Pa,Pb)),

allen(Rel, PrjSt,PrjLm).

Buffer Size. In sketch maps, the distances between depicted

objects are distorted. The distances between landmarks and ad-

jacent street segments are always variant between sketch maps

of the same location. Therefore, defining a common buffer

size for all freehand sketch maps is not possible. In order

to compute relative buffer, we define an automatic method

as rule in CLP(QS) which computes minimum distances

between landmarks and street segments and then considers the

maximum distance as buffer size from the computed minimum

distances.

set_buffer_distance :-

max_landmark_street_distance(value(D)),

retractall(buffer(_)), assert(buffer(D)).

max_landmark_street_distance(value(D)) :-

setof(Dist,

L^nearest_street_distance(value(Dist), L),Dists),

sort(Dists,SDists), reverse(SDists,[D|_]).

nearest_street_distance(value(MinD), L) :-

object(L, type(landmark)), representation(L,LG),

setof(D,

S^SG^( object(S, type(street)), representation(S,SG),

distance(value(D), LG, SG) ),Ds),

sort(Ds,[MinD|_]), write(’.’),flush.

Adjacency. We define a rule for adjacency as a qualitative

constraint between landmarks and street segments. A landmark

is considered adjacent when the polygonal footprint of the

landmark intersects with the buffer of the reference street

segment. The adjacency rule is used to define linear ordering

locally, between nearby objects.

adjacent(Lm, St) :-

object(Lm, type(landmark)),

object(St, type(street)), buffer(Buf),

representation(St, StreetGeom),

representation(Lm, LandmarkGeom),

distance(value(Dist), LandmarkGeom, StreetGeom),

Dist < Buf.

Linear Projection. This spatial rule projects landmarks onto

adjacent street segments. The projections of landmarks onto

street segments are represented as pairs of intervals with

start and end points. The rule also projects spatial entities

such as point-to-line, line-to-line, and polygon-to-line objects.

The projected intervals together with the intervals of street

segments provide ordering information between spatial objects

along a route.

projection(point(Xpj, Ypj), point (Xp,Yp),

segment(Pa,Pb)) :-

projection(

value(V), point(Xp,Yp), segment(Pa,Pb)),

reconstruct_projection(

point(Xpj,Ypj), value(V), segment(Pa,Pb)).

Qualitative Interval Ordering. These spatial rules define

linear ordering relations between projected intervals. The

rule compares the start and end points of two intervals and

represents them as relations as defined in Allen’s Interval

Algebra (IA).

allen(before, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A2 < B1}.

allen(after, interval(A1,A2), interval(B1,B2)) :-

allen(before, interval(B1,B2), interval(A1,A2)).

allen(meets, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A2 =:= B1}.

allen(met_by, interval(A1,A2), interval(B1,B2)) :-

allen(meets, interval(B1,B2), interval(A1,A2)).

allen(overlaps, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A2 > B1, A1 < B1, A2 < B2}.

allen(overlapped_by, interval(A1,A2), interval(B1,B2)) :-

allen(overlaps, interval(B1,B2), interval(A1,A2)).

allen(starts, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A1 =:= B1, A2 < B2}.

allen(started_by, interval(A1,A2), interval(B1,B2)) :-

allen(starts, interval(B1,B2), interval(A1,A2)).

Fig. 3. (a) Orientation of adjacent landmarks with respect to street segments
(b) Qualitative constraints representing orientation relations.

allen(during, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A1 > B1, A2 < B2}.

allen(contains, interval(A1,A2), interval(B1,B2)) :-

allen(during, interval(B1,B2), interval(A1,A2)).

allen(finishes, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A1 > B1, A2 =:= B2}.

allen(finished_by, interval(A1,A2), interval(B1,B2)) :-

allen(finishes, interval(B1,B2), interval(A1,A2)).

allen(equal, interval(A1,A2), interval(B1,B2)) :-

{A1 < A2, B1 < B2},

{A1 =:= B1, A2 =:= B2}.

C. Rules for Relative Orientation of Landmarks

It is common to use points as basic entities in positional

reasoning [11, 19]. In [14], we investigate different qualitative

representations in order to formalize orientation information

of landmarks with respect to adjacent street segments and

relations defined in the Left-Right (LR) calculus [28]. The

LR calculus deals with point type entities in the plane R2.

It describes the position of a point C with respect to two

other points A (the origin) and B (the relatum). However,

in sketch maps landmarks are extended objects approximated

by polygons and considering the centroids of landmarks loses

shape information of the depicted objects.

We propose the following relations for representing relative

orientation of landmarks and street segments at a conceptual

level in sketch maps (see Figure 3). Our representation consists

of six binary relations between two objects: left_of , right_of ,
crosses, crossed_by, front_of , and back_of . These relations

capture key equivalence classes of spatial configurations that

are preserved in sketch maps.

Orientation Relations. These rules define six orientation

relations between depicted landmarks and street segments. A

landmark is considered to be on the left of a street segment

if all of the vertices of the polygonal representation of the

landmark are on the left side of the street segment. The

landmark crosses an adjacent street segment if some vertex

of the landmark is left_of the street segment, some vertex is

right_of the street segment, and the projection of the landmark

onto the line collinear with the street intersects the street

segment.

orientation(left_of, polygon(Pts), line(A,B)) :-

orientation_all_(left_of, Pts, line(A,B)).

orientation(crosses, polygon(Pts), line(A,B)) :-

projection(PrjLm,polygon(Pts),line(A,B)),

projection(PrjSt,segment(A,B),line(A,B)),

not(allen(before, PrjSt, PrjLm)),

not(allen(after, PrjSt, PrjLm)),

orientation_some_(left_of, Pts, line(A,B)),

orientation_some_(right_of,Pts, line(A,B)).

A landmark is in front_of an adjacent street segment if

some vertex of the landmark is on the left_of the street

segment, some vertex is on the right_of the street segment,

and the projection of the landmark onto the line collinear

with the street is not on the street segment (see Figure 3b).

The respective inverse orientation relations are: right_of ,
crossed_by and back_of .

orientation(front_of, polygon(Pts), line(A,B)) :-

orientation_some_(left_of, Pts, line(A,B)),

orientation_some_(right_of,Pts, line(A,B)),

projection(PrjLm,polygon(Pts),line(A,B)),

projection(PrjSt,segment(A,B),line(A,B)),

allen(before, PrjSt, PrjLm).

D. Rules for Cyclic Ordering as Constraints

This sketch aspect describes the angular ordering of nearby

landmarks and connected street segments as referent objects

around a street junction. Similar to linear ordering, Wang

et al. [33, 34] find the cyclic ordering of both landmarks

and street segments around reference junctions as a suitable

representation for sketch map alignment.

In [16], we proposed a coarse version of the Cyclic Interval

Algebra (CIA) [22] to formalize cyclic ordering of depicted

sketch map objects around reference junctions. The c-intervals

of landmarks and street segments are their projections onto a

central point, i.e. a reference junction. The projection is given

by sweeping the 360o view at a junction in a counter-clockwise

direction. We define spatial rules in CLP(QS) for deriving the

cyclic ordering of depicted objects. As with linear ordering,

the spatial rule is used to generate a qualitative constraint

network between landmarks and street segments based on

adjacency, cyclic projections, and qualitative relations between

cyclic intervals.

Adjacency Buffer Size. Analogous to the computation of

buffer size around a street segment, this rule defines the buffer

size around a reference object. The buffer size around junc-

tions is used to define adjacency relations between landmarks

and reference junctions.

Fig. 4. (a) Cyclic ordering of adjacent landmarks with respect to junction
B (b) qualitative constraints representing cyclic ordering relations

set_buffer_distance :-

max_landmark_junction_distance(value(D)),

retractall(buffer(_)), assert(buffer(D)).

max_landmark_junction_distance(value(D)) :-

setof(Dist,

L^nearest_junction_distance(value(Dist), L),Dists),

sort(Dists,SDists), reverse(SDists,[D|_]).

nearest_junction_distance(value(MinD), L) :-

object(L, type(landmark)), representation(L,LG),

setof(D, S^SG^(

object(S, type(junction)),

representation(S,SG),

distance(value(D), LG, SG)

),Ds),

sort(Ds,[MinD|_]).

Cyclic Projection. This rule projects the adjacent start and end

points of landmarks onto reference junctions. The projected

points represent c-intervals of adjacent landmarks. In order to

further constrain the relative positions of landmarks around

junctions, we also consider the c-intervals of street segments

connected at the reference junctions. Each projected point in

a c-interval represents a pair of connected street segments at a

junction. The relations between these intervals represents the

cyclic ordering of connected street segments.

cyclic_ordering(Rel, Lm1, Lm2, Jun) :-

object(Jun, type(junction)),

object(Lm1, type(landmark)),

object(Lm2, type(landmark)),

representation(Lm1,GLm1),

representation(Lm2,GLm2),

representation(Jun,GJun),

cyclic_projection(PrjLm1, GLm1, GJun ),

cyclic_projection(PrjLm2, GLm2, GJun),

cyclic_ordering(Rel, PrjLm1,PrjLm2).

Qualitative Cyclic Ordering. These rules define the cyclic

ordering of landmarks and connected street segments using the

projected c-intervals. It captures a panoramic view of objects

around junctions in the form of cyclic ordering constraints.

As illustrated in Figure 5, cyclic intervals are represented as

a centre angle θ (measured counter-clockwise from the origin

direction (1, 0)) and an angular half-distance h between the

centre and the end-points of the cyclic interval.1 Figure 4

illustrates the disconnected (dc), during (d), met-by (mi), and

meets (m) cyclic relations between two c-intervals of objects

around a junction.

1The half-distance of an interval is analogous to the radius of a circle being
half of the circle’s diameter.

cyclic origin (1,0)

landmark

junction

projected
cyclic interval

h
h

ϑ

Fig. 5. Projected cyclic interval of a polygonal landmark onto a junction
point. The cyclic interval is represented as a centre angle θ (relative to the
angular origin (1, 0)) and angular half-distance h.

cyclic_ordering(disconnected,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

distance(radian(Dab),

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{Dab > Ha + Hb}.

cyclic_ordering(externally_connects,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

distance(radian(Dab),

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{Dab =:= Ha + Hb}.

cyclic_ordering(partially_overlaps,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

distance (radian(Dab),

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{Dab < (Ha + Hb), Dab > abs(Ha - Hb)}.

cyclic_ordering(tangential_proper_part,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

distance (radian(Dab),

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{Dab =:= (Hb - Ha), Ha < Hb}.

cyclic_ordering(nontangential_proper_part,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

distance (radian(Dab),

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{Dab < (Hb - Ha), Ha < Hb}.

cyclic_ordering(meets,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

cyclic_ordering(externally_connects,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)),

{(Ca + Ha) modulo 2*pi =:= (Cb - Hb) modulo 2*pi}.

cyclic_ordering(met_by,

cyclic_interval(centre(Ca),Ha),

cyclic_interval(centre(Cb),Hb)) :-

cyclic_ordering(meets,

cyclic_interval(centre(Cb),Hb),

cyclic_interval(centre(Ca),Ha)).

V. DISCUSSION AND USE CASE

In this section we discuss preliminary results of using our

representations for matching. We have conducted a simple

pilot study to investigate the ways in which various represen-

tations can be combined and utilised for the task of matching

(i.e. aligning) spatial objects from sketch maps with geo-

referenced maps.

The objective of the alignment task is to accurately match

objects from the sketch map with objects in the geo-referenced

map according to their relevant qualitative spatial relations.

This is a challenging task in general [27], and we are investi-

gating the use of a “mosaic” of different representations that

can be combined to increase the accuracy of the matching pro-

cess, e.g. a combination of topological relations between city

blocks, linear representations between landmarks along street

segments, cyclic representations between landmark around

junctions, and so on.

In this simple pilot study we explore the combination of

topological relations between city blocks and linear relations

between landmarks based on a real, user-generated sketch map

of Münster, Germany (see Figures 6 and 7). That is, we firstly

match the maps based on city blocks using topological rela-

tions (primarily adjacency), as reported in [17]. As we report

in [17], city blocks can often be matched accurately based

on topological information. Figure 8 illustrates the extract of

the sketch map (Figure 6) and geo-referenced map (Figure 7)

comparing landmarks within a correctly matched city block.

We employ our linear representation to match landmarks

within each city block. Landmarks are linearly ordered based

on their projection onto the streets that define the city block.

Figure 8 illustrates the derived ordering of the landmarks; for

clarity, only the upper horizontal street in the sketch map is

shown. The light grey region indicates the proximity threshold

used to define adjacency of landmarks with respect to street

segments. Projection of the landmark is based on the region of

the landmark within the proximity threshold. Based on these

projections, landmarks r2, r3, r4 maintain a similar ordering

in both the sketch and geo-referenced maps, thus assisting in

the matching between landmark objects within a city block.

This pilots study also highlights the complexity of the

matching task: observe that landmark r1 was not sketched

by the participant (i.e. r1 is missing from the sketch map,

Figure 8(a)). Moreover, according to our projection represen-

tation in the geo-referenced map (Figure 8(b)), r1 overlaps

r2 in the first street of the geo-referenced map, but is also

projected onto the last street that defines the block. We need

to formalise further spatial domain knowledge in order for a

matching algorithm to correctly interpret such combinations

of qualitative relations between landmarks along a path.

The evaluation of proposed representations based on qualita-

tive matching of sketch maps with geo-referenced maps is on-

going research work. In future work we are also investigating

alternative landmark projections. For example, the front face

of the polygon may be better indication of how people perceive

the ordering information of landmarks along a particular route,

Fig. 6. Sketch map of a region in Münster, Germany.

Fig. 7. Geo-referenced map of Münster, Germany.

rather than the projection of the entire polygon within the

proximity threshold.

VI. CONCLUSIONS

In this paper we introduce a set of spatial relations that

formalize key equivalence classes of spatial configurations

preserved in sketch maps; the relations have been defined

using the declarative spatial reasoning system CLP(QS). Our

rules address: (1) the linear ordering of landmarks and street

segments along any route; (2) the cyclic ordering of landmarks

and connected street segments around reference junctions;

(3) the orientation of landmarks with respect to nearby street

segments. The rules also define an appropriate buffer size in

sketch maps using the minimum distances between landmarks

and street segments. The buffer size is used to define proximity

and adjacency relations as qualitative relations between nearby

objects.

We derive ordering and relative orientation relations as sets

of Prolog facts using CLP(QS). We then use these facts to

generate Qualitative Constraint Networks (QCNs) of sketch

and geo-referenced maps. These qualitative networks are used

in a range of tasks, for example (a) to provide a high-level

qualitative query interface to both sketched and geo-referenced

maps, and (b) to facilitate matching sketched maps to geo-

referenced maps i.e. aligning qualitative spatial information in

sketch maps with corresponding spatial information in geo-

referenced maps. The evaluation of proposed representations

based on qualitative matching of sketch maps with geo-

referenced maps is ongoing research work.

Fig. 8. Deriving linear ordering relations of projected landmarks per block in
the sketch and geo-referenced map; only the projection of adjacent landmarks
onto the upper horizontal street is illustrated.

